
Soundex
SYNOPSIS

This pracnique calculates the Soundex code of a word.

Language: Ada

Compiler: gnatmake

Experience Level: intermediate

Skills: program design, loops, functions, arrays, strings, unbounded strings

INTRODUCTION 
The Soundex Code, was developed by Robert C. Russell of Pittsburgh, Pennsylvania in the early 1900s 

(U.S. patent 1,261,167, April 2, 1918). It is a phonetic algorithm which provides a way of reducing similar 
sounding words to a common code. Soundex can therefore help find a surname in census documents 
and surname databases, even though it may have been recorded under various spellings. The algorithm 

drops vowels and silent letters, removes double letters and then assigns the remaining letters numbers 
from the following classes:

 Letters  Code  Description
 BFPV  1  Labial

 CGJKQSXZ 2  Gutterals and sibilants
 DT  3  Dental

 L  4  Long liquid
 MN  5  Nasal
 R  6  Short liquid

If the Soundex codes of two words are the same, that means they are pronounced the same way. Words 

are first abbreviated using the following simple rules:

 Leave the first letter of the word as is.

 Delete all occurrences of A, E, I O, U, W, H, and Y.
 Replace double letters by a single letter.

 Stop after 4 characters are produced.

Improvements to Soundex are the basis for many modern phonetic algorithms. For example:

 HAWKSLEY → HAWKSLEY → HKSL → H224

There are several special cases when calculating a soundex code:
Letters with the same soundex number that are immediately next to each other are discarded. So Pfizer 

becomes Pizer, Sack becomes Sac, Czar becomes Car, Collins becomes Colins, and Mroczak becomes 
Mrocak. If two letters with the same soundex number separated by "H" or "W", only use the first letter. 

So Ashcroft is treated as Ashroft.

Here are some examples: 

perl, pearl, peril, prowl, puerile - P64

Smyth, Smithe, Smithee, Smith - S530
Lukasiewicz, Lissajous - L222

THE MAIN CODE 
This program could be written in C, however it would be longer because of the inherent lack of some 

utilities that make life easier. Although strings in Ada can be challenging, the use of both string and 
unbounded_string types, and built in string functions makes processing strings a reasonable task. This 
program will read in two words, and decide if they “sound” the same (not implementing the special 

cases).

The first piece of code to create deals with reading in the two words from the standard input.

   put_line("Enter two words to compare: ");
   get_line(word1);
   get_line(word2);

Both word1 and word2 are declared as unbounded_string ‘s which means that they are not restricted in 

size. If we create a “normal” string of length 10:

   word : string(1..10);

Then that string must always be of length 10 - we can’t input a string of only length 4 and store it in 
word. Unbounded strings use two packages: ada.Strings.Unbounded and 
ada.Strings.Unbounded.Text_IO. 

Ideally to properly process these words, they would both be in lowercase (it just makes the soundex 

code easier). This can be achieved using the function translate() found in the package 
ada.Strings.Maps.constants. In this case we use the clause Lower_Case_Map to achieve the mapping 
to lowercase:

   translate(word1, Lower_Case_Map);
   translate(word2, Lower_Case_Map);

Now we pass both words to a user-defined function findSoundex(), output the soundex strings for each, 
and decide whether the words “sound” the same.

   findSoundex(word1,sndx1);   
   findSoundex(word2,sndx2);
   
   put_line(word1 & " soundex = " & sndx1);
   put_line(word2 & " soundex = " & sndx2);
   new_line;
   
   if sndx1 = sndx2 then
       put_line("Words sound the same.");
   else
       put_line("Words sound different.");   
   end if;

Notice that the put_line() function uses the & to concatenate substrings together, which is easier than 
writing separate statements. 

THE PROCEDURE findSoundex()
This program essentially has one other procedure, which derives the soundex of a word. Here is what the 
basic procedure looks like (a procedure does not return anything).

    procedure findSoundex(strWord: in unbounded_string; 
                          sndx: out unbounded_string) is
       i, j, sL: integer;
    begin

    end findSoundex;

The function has two parameters: strWord, which is the unbounded_string that is input, and sndx which 
is the unbounded_string to be output. Deriving the program, really just involves implementing the 

soundex rules. Now whilst an unbound_string is ideal because of its unrestricted length, it does suffer 
from issues when it comes to accessing and replacing elements (it’s just slightly more tedious).

The first thing to do is determine the length of the unbounded string strWord:

 sL := length(strWord);

As the soundex requires the first letter of the word to be kept intact, we can also perform this task:

 append(sndx, to_upper(element(strWord,1)));

This code extracts element 1 from strWord, and appends it to the unbounded string sndx. To make it 
consistent with normal soundex output, the first character is converted to uppercase, using the function 
to_upper() (found in ada.Characters.Handling) The append() function is used to add an element to an 

unbounded string (as none exist initially). The word in strWord can then be processed from element 2 to 
sL. Here is the loop which iterates though each element:

   while i <= sL and j <= 4 loop
      ch := element(strWord,i);
      if ch = element(strWord,i-1) then
         i := i + 1;     
      end if;
      -- soundex processing logic
   end loop;

The loop stops when the length of strWord is exceeded, or the length of the soundex exceeds 4 

characters (one of the specifications). The element is extracted using the function element(), which 

extracts an element from an unbounded string. If the character is the same as the previous one, then it is 

ignored, and the index i incremented (deleting occurrences of double letters). 

The soundex processing logic, is a nested if structure of the form:

   if ch not in 'a' | 'e' | 'h' | 'i' | 'o' | 'u' | 'w' | 'y' then
      if ch in 'b' | 'f' | 'p' | 'v' then
         append(sndx,'1');
         j := j + 1;
      elsif ch in 'c' | 'g' | 'j' | 'k' | 'q' | 's' | 'x' | 'z' then
         append(sndx,'2');
         j := j + 1;
      elsif ch in 'd' | 't' then
         append(sndx,'3');
         j := j + 1;
      elsif ch in 'l' then
         append(sndx,'4');
         j := j + 1;
      elsif ch in 'm' | 'n' then
         append(sndx,'5');
         j := j + 1;
      elsif ch in 'r' then
         append(sndx,'6');
         j := j + 1;
      end if;
      i := i + 1;
   else
      i := i + 1;
   end if;

The first if statement ignores all occurrences of the letter [a, e, h, i, o, u, w, y] . The inner nested if 
statement determines which code is assigned to the letters which remain. Note the use of append() to 
add the code number to the soundex. Note that the conditional expression in each portion of the nested 
if uses a membership test (Ada 2012). This is much sleeker than in C. For example this Ada code:

 if ch not in 'a' | 'e' | 'h' | 'i' | 'o' | 'u' | 'w' | 'y' then

would look like this in C:

 if (ch != 'a' || ch != 'e' || ch != 'h' || ch != 'i' || 
       ch != 'o' || ch != 'u' || ch != 'w' || ch != 'y')

Which is a bit more of a hassle to write.

TESTING THE PROGRAM 
Now compile and test the program using the test data given in the example:

Enter two words to compare:
peace
piece
peace soundex = P2
piece soundex = P2

Words sound the same.

THE WHOLE PROGRAM
Here is the whole program:

with ada.Text_IO; use Ada.Text_IO;
with ada.Strings.Unbounded; use ada.Strings.Unbounded;
with ada.Strings.Unbounded.Text_IO; use ada.Strings.Unbounded.Text_IO;
with ada.Strings.Maps.constants; use ada.Strings.Maps.Constants;
with ada.Characters.Handling; use ada.Characters.Handling;

procedure soundex is
    word1, word2, sndx1, sndx2: unbounded_string;
    
    procedure findSoundex(strWord: in unbounded_string; 
                             sndx: out unbounded_string) is
       i, j, sL: integer;
       ch: character;
    begin
       sL := length(strWord);
       append(sndx, to_upper(element(strWord,1)));
       
       j := 2;
       i := 2;
       while i <= sL and j <= 4 loop
          ch := element(strWord,i);
          if ch = element(strWord,i-1) then
             i := i + 1;     
          end if;
          if ch not in 'a' | 'e' | 'h' | 'i' | 'o' | 'u' | 'w' | 'y' then
             if ch in 'b' | 'f' | 'p' | 'v' then
                append(sndx,'1');
                j := j + 1;
             elsif ch in 'c' | 'g' | 'j' | 'k' | 'q' | 's' | 'x' | 'z' then
                append(sndx,'2');
                j := j + 1;
             elsif ch in 'd' | 't' then
                append(sndx,'3');
                j := j + 1;
             elsif ch in 'l' then
                append(sndx,'4');
                j := j + 1;
             elsif ch in 'm' | 'n' then
                append(sndx,'5');
                j := j + 1;
             elsif ch in 'r' then
                append(sndx,'6');
                j := j + 1;
             end if;
             i := i + 1;
          else
             i := i + 1;
          end if;
       end loop;   

    end findSoundex;

begin
   put_line("Enter two words to compare: ");
   get_line(word1);
   get_line(word2);
   
   translate(word1, Lower_Case_Map);
   translate(word2, Lower_Case_Map);

   findSoundex(word1,sndx1);   
   findSoundex(word2,sndx2);
   
   put_line(word1 & " soundex = " & sndx1);
   put_line(word2 & " soundex = " & sndx2);
   new_line;
   
   if sndx1 = sndx2 then
       put_line("Words sound the same.");
   else
       put_line("Words sound different.");   
   end if;
end soundex;

